Detecting Income "Leaks" in General Equilibrium Models Formulated with `GAMS/MCP`

Thomas F. Rutherford
ETH Zürich

November 21, 2008

A pure-exchange market equilibrium model consists of solutions to a system of equations:

$$\xi(p) = 0$$

where $\xi : R^n \rightarrow R^n$. The market excess demand function incorporates to the optimizing responses of multiple households, h, i.e.

$$\xi(p) = \sum_h \omega_h - d_h(p)$$

in which $d_h(p)$ is the uncompensated demand function, a solution to the utility maximization problem:

$$\max_x U_h(x)$$

s.t.

$$p^T x = p^T \omega_h$$

The market price vector appears in the linear constraint, hence the optimal demand vector is unaffected by scaling of the price level – the demand functions are all homogeneous of degree *zero* in prices, i.e.:

$$\xi(\lambda p) = \lambda^0 \xi(p) = \xi(p).$$

Linear dependence of the system of market clearance equations can alternatively be attributed to *Walras law.* Given budget constraints for each of the individual choice problems, we have:

$$p^T \xi(p) = 0 \quad \forall p$$

Both of these conditions confirm that the system of equations describing the economic equilibrium is over-determined. If any $n-1$ of these values is satisfied for a given prices vector p^*, then the nth equation is also automatically satisfied, provided that the demand functions are correctly specified.

In the GAMS/MCP framework variables are associated with equations, and whenever a variable is fixed the corresponding equation is omitted. The two

Table 1: Equilibrium Prices for Alternative Numeraires

	Numeraire Good Valid Model			Numeraire Good Invalid Model		
	G1	G2	G3	G1	G2	G3
G1	1.000	0.821	0.864	1.000	1.068	1.035
G2	1.218	1.000	1.052	1.434	1.000	1.287
G3	1.158	0.950	1.000	1.334	1.235	1.000
WALRAS	3.47E-12	-5.27-9	-1.21-6	-0.63	-0.63	-0.63

GAMS programs attached to this note illustrate this idea. A random exchange model is generated, and the solution is computed for each of n numeraire prices. In each case, we report the resulting equilibrium prices and the computed deviation in the omitted market. (The two programs are identical apart from their representation of the equilibrium conditions. The second program illustrates the new $macro statement which will be available with GAMS version 22.9.)

As illustrated in the left side of Table 1, when the exchange model is properly specified, the same equilibrium prices are return irrespective of the numeraire specification. Likewise the "Walras check", the imbalance in the omitted numeraire market clearance condition is zero when the model has no income "leaks". Conversely, when an imbalance is introduced in the model, both absolute and relative prices depend on the numeraire choice and the "Walras check" is nonzero, as indicted on the right side of Table 1.

```
$title  General Equilibrium Exchange Model -- Conventional GAMS Syntax

set     h       Households /h1*h3/
        g       Goods /g1*g3/;

parameter       theta(g,h)      Share parameters,
                sigma(h)        Substitution parameters,
                omega(g,h)      Endowment parameters,
                bug(h)          Vector of spurious income entries;

theta(g,h) = uniform(0,1);
alias (g,gg);
theta(g,h) = theta(g,h)/sum(gg,theta(gg,h));
sigma(h) = uniform(0,3);
omega(g,h) = uniform(0,1);
bug(h) = 0;

VARIABLE        P(g)    Market price for good g,
                Y(h)    Household income,
                C(h)    Unit cost of consumption,
                D(g,h)  Uncompensated demand
                XI(g)   Market excess demand;

EQUATION        income(h), cost(h), demand(g,h), market(g), equil(g);

income(h)..     Y(h) =e= sum(g, P(g)*omega(g,h)) + bug(h);

cost(h)..       C(h) =e= sum(g, theta(g,h) * P(g)**(1-sigma(h)))**(1/(1-sigma(h)));

demand(g,h)..   D(g,h) =e= Y(h)/C(h) * theta(g,h) * (C(h)/P(g))**sigma(h);

market(g)..     XI(g) =e= sum(h, D(g,h)) - sum(h, omega(g,h));

equil(g)..      -XI(g) =e= 0;

model edgeworth /income.Y, cost.C, demand.D, market.XI, equil.P /;

P.L(g) = 1;
Y.L(h) = 1;
C.L(h) = 1;

parameter       equilp  Equilibrium prices with alternative numeraires;
loop(gg,

*       Remove upper and lower bounds for all prices:

        P.UP(g) = +inf;         P.LO(g) = 0;

*       Fix one price index as numeraire:
```

```
        P.FX(gg) = 1;

        solve edgeworth using mcp;

        equilp(g,gg) = P.L(g);
        equilp("Walras",gg) = P.M(gg);
);
display "Equilibrium prices and cross check for consistent model:",equilp;

*       Introduce some errors in the calculations:

bug(h) = uniform(0,1);


loop(gg,

*       Remove upper and lower bounds for all prices:

        P.UP(g) = +inf;         P.LO(g) = 0;

*       Fix one price index as numeraire:

        P.FX(gg) = 1;

        solve edgeworth using mcp;

        equilp(g,gg) = P.L(g);
        equilp("Walras",gg) = P.M(gg);
);
display "Equilibrium prices and cross check for inconsistent model:",equilp;
```

```
$title  General Equilibrium Exchange Model -- Using GAMS 22.9 $macro Definitions

set     h       Households /h1*h3/
        g       Goods /g1*g3/;

parameter       theta(g,h)      Share parameters,
                sigma(h)        Substitution parameters,
                omega(g,h)      Endowment parameters,
                bug(h)          Vector of spurious income entries;

theta(g,h) = uniform(0,1);
alias (g,gg);
theta(g,h) = theta(g,h)/sum(gg,theta(gg,h));
sigma(h) = uniform(0,3);
omega(g,h) = uniform(0,1);
bug(h) = 0;

VARIABLE        P(g)            Market price for good g;

*       Define household income, cost index and uncompensated
*       demand functions:

$macro  Y(h)    (sum(g.local, P(g)*omega(g,h)) + bug(h))
$macro  C(h)    (sum(g.local, theta(g,h) * P(g)**(1-sigma(h)))**(1/(1-sigma(h))))
$macro  D(g,h)  (Y(h)/C(h) * theta(g,h) * (C(h)/P(g))**sigma(h))
$macro  XI(g)   sum(h, D(g,h) - omega(g,h))

EQUATION        market(g)       Market clearance for good g;

market(g)..     -XI(g) =e= 0;

model edgeworth /market.P/;

P.L(g) = 1;

parameter       equilp  Equilibrium prices with alternative numeraires;

loop(gg,

*       Remove upper and lower bounds for all prices:

        P.UP(g) = +inf;         P.LO(g) = 0;

*       Fix one price index as numeraire:

        P.FX(gg) = 1;

        solve edgeworth using mcp;

        equilp(g,gg) = P.L(g);
```

```
        equilp("Walras",gg) = P.M(gg);

);

display "Equilibrium prices and cross check for consistent model:",equilp;

*       Introduce some errors in the calculations:

bug(h) = uniform(0,1);


loop(gg,

*       Remove upper and lower bounds for all prices:

        P.UP(g) = +inf;         P.LO(g) = 0;

*       Fix one price index as numeraire:

        P.FX(gg) = 1;

        solve edgeworth using mcp;

        equilp(g,gg) = P.L(g);
        equilp("Walras",gg) = P.M(gg);
);
display "Equilibrium prices and cross check for inconsistent model:",equilp;
```